Automating the Implicit Function Theorem

```
x = 0; Remove["Global`*"]
```

Introduction

Economics revolves around solving equilibrium models

Common features of equilibrium models
(1) They are often solutions to systems of equations of analytic functions
(2) Different models often have common structure in terms of functional forms, and differ only in terms of parameter values
(3) There are usually a few special (often degenerate) cases where we can solve for the solution explicitly
(4) The implicit function theorem tells us that there is an analytic map between exogenous parameters and the equilibrium outcome near these cases.

Our objective today:
(1) Display the structure of basic economic models, using simple examples
(2) Show how one can use AD and IFT to compute equilibria for pieces of the parameter space
(3) Argue that for many purposes this approach may dominate numerical methods for solving specific instances

A Specific Example - Easy one

Let's examine a very simple example. Suppose that p is the price of a good and the demand function for that good is

In[1]:=
```
Dmd[p_] = p^-3;
```

Suppose that producers pay a tax of τ p for each unit it sells where τ is the tax rate (like a VAT) and that supply is a function of the after-tax price received by the producer

In[2]:=
```
Supply[p_, τ_] = (p (1 - τ))^(1/2);
```

The excess demand for price p and tax rate τ is

In[3]:=
```
ExDmd[p_, τ_] = Dmd[p] - Supply[p, τ]
```

Out[3]= $\frac{1}{p^3} - \sqrt{p\,(1-\tau)}$

The true solution is

In[4]:=
```
Ptrue[τ_] = (1 - τ)^(-1/7);
```

Examination of the excess demand function shows that price is 1 when the tax is τ=0. Therefore

In[5]:=
```
P[0] = 1;
```

Write the excess demand as a function of the tax τ and the equilibrium price P[τ]

```
In[6]:= ExDmdTax[τ_] = ExDmd[P[τ], τ]
```

Out[6]= $\frac{1}{P[\tau]^3} - \sqrt{(1-\tau)\, P[\tau]}$

ExDmdTax[τ]=0 for all tax rates τ.

The task: Use the parameterized equilibrium equation, the P[0]=1 condition, and the IFT to trace out the equilibrium manifold, P[τ], for τ close to zero.

Differentiate ExDmdTax[τ] at τ=0 to compute P'[0]

```
ExDmdTax'[0] // Expand
```

$\frac{1}{2} - \frac{7\, P'[0]}{2}$

```
solp1 = Solve[% == 0, P'[0]][[1]]
```

$\left\{P'[0] \to \frac{1}{7}\right\}$

We do the same to get P''[0] and P'''[0].

```
Solve[ExDmdTax''[0] == 0, P''[0]][[1]]
```

$$\left\{P''[0] \to \frac{1}{14}\left(1 + 2\,P'[0] + 49\,P'[0]^2\right)\right\}$$

```
solp2 = % //. solp1
```

$$\left\{P''[0] \to \frac{8}{49}\right\}$$

```
Solve[ExDmdTax'''[0] == 0, P'''[0]][[1]];
solp3 = % //. solp1 //. solp2
```

$$\left\{P^{(3)}[0] \to \frac{120}{343}\right\}$$

The degree 3 Taylor series of P[τ] at τ=0 is

```
Series[P[τ], {τ, 0, 3}]
% /. solp3
% /. solp2
% /. solp1
% // Normal
```

$$1 + P'[0]\,\tau + \frac{1}{2} P''[0]\,\tau^2 + \frac{1}{6} P^{(3)}[0]\,\tau^3 + O[\tau]^4$$

$$1 + P'[0]\,\tau + \frac{1}{2} P''[0]\,\tau^2 + \frac{20\,\tau^3}{343} + O[\tau]^4$$

$$1 + P'[0]\,\tau + \frac{4\,\tau^2}{49} + \frac{20\,\tau^3}{343} + O[\tau]^4$$

$$1 + \frac{\tau}{7} + \frac{4\,\tau^2}{49} + \frac{20\,\tau^3}{343} + O[\tau]^4$$

$$1 + \frac{\tau}{7} + \frac{4\,\tau^2}{49} + \frac{20\,\tau^3}{343}$$

SolveAlways Command

The degree 3 Taylor series of ExDmdTax[τ] at τ=0 is

```
ser = Series[ExDmdTax[τ], {τ, 0, 3}]
```

$$\left(\frac{1}{2}-\frac{7\,P'[0]}{2}\right)\tau+\frac{1}{8}\left(1+2\,P'[0]+49\,P'[0]^2-14\,P''[0]\right)\tau^2+\frac{1}{48}\left(3+3\,P'[0]-3\,P'[0]^2-483\,P'[0]^3+6\,P''[0]+294\,P'[0]\,P''[0]-28\,P^{(3)}[0]\right)\tau^3+O[\tau]^4$$

We will use the SolveAlways command to solve out for the derivatives of P at τ=0.

```
sol = SolveAlways[ser == 0, τ][[1]]
```

$$\left\{P^{(3)}[0]\to\frac{120}{343},\ P''[0]\to\frac{8}{49},\ P'[0]\to\frac{1}{7}\right\}$$

We now substitute this into the the Taylor series expansion for P[τ] at τ=0.

```
Series[P[τ], {τ, 0, 3}] /. sol
```

$$1+\frac{\tau}{7}+\frac{4\,\tau^2}{49}+\frac{20\,\tau^3}{343}+O[\tau]^4$$

The degree 3 Taylor series of P[τ] at τ=0 is

```
solPp[τ_] = Series[P[τ], {τ, 0, 3}] /. sol // Normal
```

$$1+\frac{\tau}{7}+\frac{4\,\tau^2}{49}+\frac{20\,\tau^3}{343}$$

■ Quality check

Question: How good is our Taylor series approximation.

Answer 1: Since we know the solution, display the relative error of the approximation. The error is less than 0.1 per cent for tax rates below 0.30; not bad. Higher order series will do better.

```
Plot[Log[10, 1 - solPp[τ] / Ptrue[τ]], {τ, 0, 1}]
```



But we generally do not know the answer!

Answer 2: Check the residual. We can always do that.

```
Residual[τ_] = ExDmdTax[τ] /. P → solPp
```

$$\frac{1}{\left(1+\frac{\tau}{7}+\frac{4\,\tau^2}{49}+\frac{20\,\tau^3}{343}\right)^3}-\sqrt{(1-\tau)\left(1+\frac{\tau}{7}+\frac{4\,\tau^2}{49}+\frac{20\,\tau^3}{343}\right)}$$

```
Plot[Log[10, Residual[τ] / Dmd[solPp[τ]]], {τ, 0, 1}]
```



The residual, normalized relative to demand, is less than 0.1 percent for tax rates below 0.30.

In practice, we check residual to determine the order of the Taylor series necessary for a good solution.

An Abstract Example in General Style

In general, we will want to compute the approximation for many different parameter sets.
Therefore, we take the following approach.

Consider a constant elasticity specification, a generalization of our simple example

```
Dmd[p_] = A p^-ν; Supply[p_, τ_] = B (p (1 - τ))^η;
ExDmd[p_, τ_] = Dmd[p] - Supply[p, τ];
ExDmdTax[τ_] = ExDmd[P[τ], τ]
```

```
A P[τ]^-ν - B ((1 - τ) P[τ])^η
```

We know the solution at τ=0:

```
P[0] = (A / B)^(1/(η+ν));
```

■ Compute Derivatives

We first compute derivatives of ExDmdTax[t] at t=0 in symbolic form

```
ExDmdTax'[0]
% // PowerExpand
% // Simplify
```

$$-A\left(\left(\frac{A}{B}\right)^{\frac{1}{\eta+\nu}}\right)^{-1-\nu}\nu P'[0]-\left(\left(\frac{A}{B}\right)^{\frac{1}{\eta+\nu}}\right)^{-1+\eta}B\,\eta\left(-\left(\frac{A}{B}\right)^{\frac{1}{\eta+\nu}}+P'[0]\right)$$

$$-A^{1+\frac{-1-\nu}{\eta+\nu}}B^{-\frac{-1-\nu}{\eta+\nu}}\nu P'[0]-A^{\frac{-1+\eta}{\eta+\nu}}B^{1-\frac{-1+\eta}{\eta+\nu}}\eta\left(-A^{\frac{1}{\eta+\nu}}B^{-\frac{1}{\eta+\nu}}+P'[0]\right)$$

$$A^{-\frac{1+\nu}{\eta+\nu}}B^{\frac{\nu}{\eta+\nu}}\left(A^{\frac{1+\eta+\nu}{\eta+\nu}}\eta-A\,B^{\frac{1}{\eta+\nu}}(\eta+\nu)\,P'[0]\right)$$

```
ExDmdTax''[0] // PowerExpand // Simplify
```

$$A^{-\frac{2+\nu}{\eta+\nu}}B^{-\frac{\eta}{\eta+\nu}}\left(-A^{\frac{2+\eta+\nu}{\eta+\nu}}B(-1+\eta)\eta-\right.$$
$$\left.A\,B^{1+\frac{2}{\eta+\nu}}\left(-\eta+\eta^2-\nu(1+\nu)\right)P'[0]^2+A^{\frac{1+\eta+\nu}{\eta+\nu}}B^{1+\frac{1}{\eta+\nu}}\eta\left(2\eta P'[0]-P''[0]\right)-A^{1+\frac{1}{\eta+\nu}}B^{\frac{1+\eta+\nu}{\eta+\nu}}\nu P''[0]\right)$$

```
ExDmdTax'''[0] // PowerExpand // Simplify
```

$$A^{\frac{-3+\eta}{\eta+\nu}} B^{\frac{\nu}{\eta+\nu}} \left(-B^{\frac{3}{\eta+\nu}} \nu (1+\nu)(2+\nu) P'[0]^3 + (-2+\eta)(-1+\eta)\eta \left(A^{\frac{1}{\eta+\nu}} - B^{\frac{1}{\eta+\nu}} P'[0]\right)^3 - 3 A^{\frac{1}{\eta+\nu}} B^{\frac{1}{\eta+\nu}} (-1+\eta)\eta \left(A^{\frac{1}{\eta+\nu}} - B^{\frac{1}{\eta+\nu}} P'[0]\right)(2 P'[0] - P''[0]) + 3 A^{\frac{1}{\eta+\nu}} B^{\frac{2}{\eta+\nu}} \nu (1+\nu) P'[0] P''[0] + A^{\frac{2}{\eta+\nu}} B^{\frac{1}{\eta+\nu}} \eta \left(3 P''[0] - P^{(3)}[0]\right) - A^{\frac{2}{\eta+\nu}} B^{\frac{1}{\eta+\nu}} \nu P^{(3)}[0]\right)$$

Collect and store these expressions.

```
derivs = {ExDmdTax'[0], ExDmdTax''[0], ExDmdTax'''[0]} // PowerExpand // Simplify;
```

■ Using Series to get derivatives

Sometimes, the following is a faster way to get these derivatives

```
ser = Series[ExDmdTax[τ], {τ, 0, 3}] // Normal;

ser = ser // PowerExpand // Simplify
```

$$A^{-\frac{1+\nu}{\eta+\nu}} B^{\frac{\nu}{\eta+\nu}} \tau \left(A^{\frac{1+\eta+\nu}{\eta+\nu}} \eta - A B^{\frac{1}{\eta+\nu}} (\eta+\nu) P'[0]\right) -$$

$$\frac{1}{2} A^{-\frac{2+\nu}{\eta+\nu}} B^{-\frac{\eta}{\eta+\nu}} \tau^2 \left(A^{\frac{2+\eta+\nu}{\eta+\nu}} B (-1+\eta) \eta + A B^{1+\frac{2}{\eta+\nu}} \left(-\eta+\eta^2-\nu(1+\nu)\right) P'[0]^2 - A^{\frac{1+\eta+\nu}{\eta+\nu}} B^{1+\frac{1}{\eta+\nu}} \eta (2 \eta P'[0] - P''[0]) + A^{1+\frac{1}{\eta+\nu}} B^{\frac{1+\eta+\nu}{\eta+\nu}} \nu P''[0]\right) +$$

$$\frac{1}{6} A^{\frac{-3+\eta}{\eta+\nu}} B^{\frac{\nu}{\eta+\nu}} \tau^3 \left(-B^{\frac{1}{\eta+\nu}} \nu \left(B^{\frac{2}{\eta+\nu}} \left(2+3\nu+\nu^2\right) P'[0]^3 - 3 A^{\frac{1}{\eta+\nu}} B^{\frac{1}{\eta+\nu}} (1+\nu) P'[0] P''[0] + A^{\frac{2}{\eta+\nu}} P^{(3)}[0]\right) + \eta \left(A^{\frac{3}{\eta+\nu}} \left(2-3\eta+\eta^2\right) - B^{\frac{3}{\eta+\nu}} \left(2-3\eta+\eta^2\right) P'[0]^3 + 3 A^{\frac{1}{\eta+\nu}} B^{\frac{2}{\eta+\nu}} (-1+\eta) P'[0] (\eta P'[0] - P''[0]) - A^{\frac{2}{\eta+\nu}} B^{\frac{1}{\eta+\nu}} \left(3\eta^2 P'[0] - 3\eta (P'[0] + P''[0]) + P^{(3)}[0]\right)\right)\right)$$

```
coeffs = CoefficientList[ser, τ]
```

$$\Big\{0,\ A^{-\frac{1+\nu}{\eta+\nu}}\, B^{\frac{\nu}{\eta+\nu}} \left(A^{\frac{1+\eta+\nu}{\eta+\nu}}\, \eta - A\, B^{\frac{1}{\eta+\nu}}\, (\eta+\nu)\, P'[0]\right),$$

$$-\frac{1}{2} A^{-\frac{2+\nu}{\eta+\nu}}\, B^{-\frac{\eta}{\eta+\nu}} \left(A^{\frac{2+\eta+\nu}{\eta+\nu}}\, B\, (-1+\eta)\, \eta + A\, B^{1+\frac{2}{\eta+\nu}} \left(-\eta+\eta^2-\nu\,(1+\nu)\right) P'[0]^2 - A^{\frac{1+\eta+\nu}{\eta+\nu}}\, B^{1+\frac{1}{\eta+\nu}}\, \eta\, (2\,\eta\, P'[0] - P''[0]) + A^{1+\frac{1}{\eta+\nu}}\, B^{\frac{1+\eta+\nu}{\eta+\nu}}\, \nu\, P''[0]\right),$$

$$\frac{1}{6} A^{\frac{-3+\eta}{\eta+\nu}}\, B^{\frac{\nu}{\eta+\nu}} \left(-B^{\frac{1}{\eta+\nu}}\, \nu \left(B^{\frac{2}{\eta+\nu}} \left(2+3\,\nu+\nu^2\right) P'[0]^3 - 3\, A^{\frac{1}{\eta+\nu}}\, B^{\frac{1}{\eta+\nu}}\, (1+\nu)\, P'[0]\, P''[0] + A^{\frac{2}{\eta+\nu}}\, P^{(3)}[0]\right) + \eta \left(A^{\frac{3}{\eta+\nu}} \left(2-3\,\eta+\eta^2\right) - B^{\frac{3}{\eta+\nu}} \left(2-3\,\eta+\eta^2\right) P'[0]^3 + 3\, A^{\frac{1}{\eta+\nu}}\, B^{\frac{2}{\eta+\nu}}\, (-1+\eta)\, P'[0]\, (\eta\, P'[0] - P''[0]) - A^{\frac{2}{\eta+\nu}}\, B^{\frac{1}{\eta+\nu}} \left(3\,\eta^2\, P'[0] - 3\,\eta\, (P'[0] + P''[0]) + P^{(3)}[0]\right)\right)\right)\Big\}$$

derivs = Rest[coeffs]

$$\Big\{A^{-\frac{1+\nu}{\eta+\nu}}\,B^{\frac{\nu}{\eta+\nu}}\left(A^{\frac{1+\eta+\nu}{\eta+\nu}}\,\eta-A\,B^{\frac{1}{\eta+\nu}}\,(\eta+\nu)\,P'[0]\right),$$

$$-\frac{1}{2}\,A^{-\frac{2+\nu}{\eta+\nu}}\,B^{-\frac{\eta}{\eta+\nu}}\left(A^{\frac{2+\eta+\nu}{\eta+\nu}}\,B\,(-1+\eta)\,\eta+A\,B^{1+\frac{2}{\eta+\nu}}\left(-\eta+\eta^2-\nu\,(1+\nu)\right)P'[0]^2-\right.$$
$$\left.A^{\frac{1+\eta+\nu}{\eta+\nu}}\,B^{1+\frac{1}{\eta+\nu}}\,\eta\,(2\,\eta\,P'[0]-P''[0])+A^{1+\frac{1}{\eta+\nu}}\,B^{\frac{1+\eta+\nu}{\eta+\nu}}\,\nu\,P''[0]\right),$$

$$\frac{1}{6}\,A^{\frac{-3+\eta}{\eta+\nu}}\,B^{\frac{\nu}{\eta+\nu}}\left(-B^{\frac{1}{\eta+\nu}}\,\nu\left(B^{\frac{2}{\eta+\nu}}\left(2+3\,\nu+\nu^2\right)P'[0]^3-3\,A^{\frac{1}{\eta+\nu}}\,B^{\frac{1}{\eta+\nu}}\,(1+\nu)\,P'[0]\,P''[0]+A^{\frac{2}{\eta+\nu}}\,P^{(3)}[0]\right)+\right.$$
$$\eta\left(A^{\frac{3}{\eta+\nu}}\left(2-3\,\eta+\eta^2\right)-B^{\frac{3}{\eta+\nu}}\left(2-3\,\eta+\eta^2\right)P'[0]^3+3\,A^{\frac{1}{\eta+\nu}}\,B^{\frac{2}{\eta+\nu}}\,(-1+\eta)\,P'[0]\,(\eta\,P'[0]-P''[0])-\right.$$
$$\left.\left.A^{\frac{2}{\eta+\nu}}\,B^{\frac{1}{\eta+\nu}}\left(3\,\eta^2\,P'[0]-3\,\eta\,(P'[0]+P''[0])+P^{(3)}[0]\right)\right)\right)\Big\}$$

■ Power Series Solution

We want to solve for the first three derivatives of P[τ] at τ=0. This is accomplished by using the derivatives for ExDmdTax at τ=0 and solving out for P'[0], P''[0], and P'''[0].

```
Pderivs = Solve[derivs == 0, {P'[0], P''[0], P'''[0]}];
```

```
% // Simplify
```

$$\left\{\left\{P^{(3)}[0] \to \frac{A^{\frac{1}{\eta+\nu}} B^{-\frac{1}{\eta+\nu}} \eta \left(6 \eta^2 + 7 \eta \nu + 2 \nu^2\right)}{(\eta+\nu)^3},\ P''[0] \to \frac{A^{\frac{1}{\eta+\nu}} B^{-\frac{1}{\eta+\nu}} \eta (2 \eta + \nu)}{(\eta+\nu)^2},\ P'[0] \to \frac{A^{\frac{1}{\eta+\nu}} B^{-\frac{1}{\eta+\nu}} \eta}{\eta+\nu}\right\}\right\}$$

Create the degree three power series for P[τ] at τ=0 using the solutions in Pderivs.

```
Series[P[τ], {τ, 0, 3}] /. Pderivs[[1]];
% // Simplify
```

$$\left(\frac{A}{B}\right)^{\frac{1}{\eta+\nu}} + \frac{A^{\frac{1}{\eta+\nu}} B^{-\frac{1}{\eta+\nu}} \eta \tau}{\eta+\nu} + \frac{A^{\frac{1}{\eta+\nu}} B^{-\frac{1}{\eta+\nu}} \eta (2 \eta + \nu) \tau^2}{2 (\eta+\nu)^2} + \frac{A^{\frac{1}{\eta+\nu}} B^{-\frac{1}{\eta+\nu}} \eta \left(6 \eta^2 + 7 \eta \nu + 2 \nu^2\right) \tau^3}{6 (\eta+\nu)^3} + O[\tau]^4$$

This is an asymptotically valid third-order solution to computing the equilibrium as we change the tax τ.

Numerical Applications

It is normally impractical to compute the abstract form of the derivatives of P. Of course, these closed-form solutions are not the objective. We generally will want to compute the Taylor series for P[τ] for some specific parameter values. The issue is when do we make those substitutions. We now follow the following strategy: compute the abstract derivatives of the implicit expression that defines P, then replace the parameters with numerical values, and solve for the derivatives of P.

Repeat the construction of the derivatives:

```
ExDmdTax'[0]
```

$$-A\left(\left(\frac{A}{B}\right)^{\frac{1}{\eta+\nu}}\right)^{-1-\nu}\nu\,P'[0]-\left(\left(\frac{A}{B}\right)^{\frac{1}{\eta+\nu}}\right)^{-1+\eta}B\,\eta\left(-\left(\frac{A}{B}\right)^{\frac{1}{\eta+\nu}}+P'[0]\right)$$

```
derivs = {ExDmdTax'[0], ExDmdTax''[0], ExDmdTax'''[0]} // PowerExpand // Simplify
```

$$\Big\{A^{-\frac{1+\nu}{\eta+\nu}}B^{\frac{\nu}{\eta+\nu}}\left(A^{\frac{1+\eta+\nu}{\eta+\nu}}\eta-A\,B^{\frac{1}{\eta+\nu}}(\eta+\nu)P'[0]\right),\ A^{-\frac{2+\nu}{\eta+\nu}}B^{-\frac{\eta}{\eta+\nu}}\Big(-A^{\frac{2+\eta+\nu}{\eta+\nu}}B(-1+\eta)\eta-$$
$$A\,B^{1+\frac{2}{\eta+\nu}}\left(-\eta+\eta^2-\nu(1+\nu)\right)P'[0]^2+A^{\frac{1+\eta+\nu}{\eta+\nu}}B^{1+\frac{1}{\eta+\nu}}\eta\,(2\,\eta\,P'[0]-P''[0])-A^{1+\frac{1}{\eta+\nu}}B^{\frac{1+\eta+\nu}{\eta+\nu}}\nu\,P''[0]\Big),$$
$$A^{\frac{-3+\eta}{\eta+\nu}}B^{\frac{\nu}{\eta+\nu}}\Big(-B^{\frac{3}{\eta+\nu}}\nu(1+\nu)(2+\nu)P'[0]^3+(-2+\eta)(-1+\eta)\eta\left(A^{\frac{1}{\eta+\nu}}-B^{\frac{1}{\eta+\nu}}P'[0]\right)^3-$$
$$3A^{\frac{1}{\eta+\nu}}B^{\frac{1}{\eta+\nu}}(-1+\eta)\eta\left(A^{\frac{1}{\eta+\nu}}-B^{\frac{1}{\eta+\nu}}P'[0]\right)(2\,P'[0]-P''[0])+$$
$$3A^{\frac{1}{\eta+\nu}}B^{\frac{2}{\eta+\nu}}\nu(1+\nu)P'[0]\,P''[0]+A^{\frac{2}{\eta+\nu}}B^{\frac{1}{\eta+\nu}}\eta\left(3\,P''[0]-P^{(3)}[0]\right)-A^{\frac{2}{\eta+\nu}}B^{\frac{1}{\eta+\nu}}\nu\,P^{(3)}[0]\Big)\Big\}$$

Then, when one wants to compute specific cases, evaluate these expressions with parameter values. This is not as easy as it sounds due to round-off error.

```
vals = Thread[{ν, η, A, B} → {2., 3., 1., 5.}];
```

In general, we need to solve the equations in a sequential linear manner.

The first derivative expression gives us a linear equation for P'[0], which we then use to solve for P'[0].

```
eq1 = derivs[[1]] /. vals // Expand
```

```
5.71096 - 13.1326 P′[0]
```

```
sol1 = Solve[eq1 == 0, P'[0]][[1]]
```

```
{P′[0] → 0.434868}
```

The second derivative implies

```
eq2 = derivs[[2]] /. vals // Expand
```

```
- 11.4219 + 47.2775 P′[0] + 0. P′[0]^2 - 13.1326 P″[0]
```

Solving for P''[0] gives us

```
sol2 = Solve[eq2 == 0, P''[0]][[1]]
```

```
{P″[0] → -0.0761462 (11.4219 - 47.2775 P′[0] + 0. P′[0]^2)}
```

which contains P'[0]. We now substitute the solution for P'[0] to complete the solution for P''[0]

```
sol2 = sol2 /. sol1
```

```
{P″[0] → 0.695788}
```

The third derivative is solved by the following sequence:

```
eq3 = derivs[[3]] /. vals // Expand
```

$$11.4219 - 141.833\, P'[0] + 195.691\, P'[0]^2 - 150.\, P'[0]^3 + 70.9163\, P''[0] - 1.35263\times 10^{-14}\, P'[0]\, P''[0] - 13.1326\, P^{(3)}[0]$$

```
sol3 = Solve[eq3 == 0, P'''[0]][[1]]
```

$$\left\{P^{(3)}[0] \to -0.0761462\,\left(-11.4219 + 141.833\, P'[0] - 195.691\, P'[0]^2 + 150.\, P'[0]^3 - 70.9163\, P''[0] + 1.35263\times 10^{-14}\, P'[0]\, P''[0]\right)\right\}$$

```
sol3 = sol3 /. sol2 /. sol1
```

$$\left\{P^{(3)}[0] \to 1.80905\right\}$$

Multiple Consumers and Firms

We represent demand and supply implicitly through optimality conditions implied by the utility and production functions:
Ui[di, p]=0 iff di is agent i's demand at price p.
Fi[qi, p(1-t)]=0 iff qi is firm i's output when after-tax price is p(1-t)

```
eqns = {U1[d1, p], U2[d2, p], U3[d3, p],
    F1[q1, p (1 - τ)], F2[q2, p (1 - τ)], (q1 + q2) - (d1 + d2 + d3)} /. p → P[τ];
eqns = % /. d1 → D1[τ] /. d2 → D2[τ] /. d3 → D3[τ] /. q1 → Q1[τ] /. q2 → Q2[τ];

eqns // TableForm
```

```
U1[D1[τ], P[τ]]
U2[D2[τ], P[τ]]
U3[D3[τ], P[τ]]
F1[Q1[τ], (1 - τ) P[τ]]
F2[Q2[τ], (1 - τ) P[τ]]
-D1[τ] - D2[τ] - D3[τ] + Q1[τ] + Q2[τ]
```

Initialize the solutions at t=0.

```
P[0] = p0;
Q1[0] = q10; Q2[0] = q20; D1[0] = d10; D2[0] = d20; D3[0] = d30;
U1[d10, p0] = 0; U2[d20, p0] = 0; U3[d30, p0] = 0;
F1[q10, p0] = 0; F2[q20, p0] = 0;
```

```
eqns0 = (D[eqns, τ]) /. τ → 0;
```

```
% // MatrixForm
```

$$\begin{pmatrix} P'[0]\, U1^{(0,1)}[d10, p0] + D1'[0]\, U1^{(1,0)}[d10, p0] \\ P'[0]\, U2^{(0,1)}[d20, p0] + D2'[0]\, U2^{(1,0)}[d20, p0] \\ P'[0]\, U3^{(0,1)}[d30, p0] + D3'[0]\, U3^{(1,0)}[d30, p0] \\ (-p0 + P'[0])\, F1^{(0,1)}[q10, p0] + Q1'[0]\, F1^{(1,0)}[q10, p0] \\ (-p0 + P'[0])\, F2^{(0,1)}[q20, p0] + Q2'[0]\, F2^{(1,0)}[q20, p0] \\ -D1'[0] - D2'[0] - D3'[0] + Q1'[0] + Q2'[0] \end{pmatrix}$$

```
vars = {P'[0], D1'[0], D2'[0], D3'[0], Q1'[0], Q2'[0]};
```

Let's use a substitution that reduces size of expressions.

```
sbs = Derivative[jj__][gg__][xx__] → Derivative[jj][gg];
```

This substitution makes our variable list simpler

```
vars /. sbs
```

```
{P', D1', D2', D3', Q1', Q2'}
```

LinSystem is the linear system implied by IFT.

```
LinSystem = CoefficientArrays[(eqns0 /. sbs), (vars /. sbs)];
```

LinSystem[[2]] is the Jacobian, and LinSystem[[1]] is the vector term.

```
jac = LinSystem[[2]];
jac // MatrixForm
```

$$\begin{pmatrix} U1^{(0,1)} & U1^{(1,0)} & 0 & 0 & 0 & 0 \\ U2^{(0,1)} & 0 & U2^{(1,0)} & 0 & 0 & 0 \\ U3^{(0,1)} & 0 & 0 & U3^{(1,0)} & 0 & 0 \\ F1^{(0,1)} & 0 & 0 & 0 & F1^{(1,0)} & 0 \\ F2^{(0,1)} & 0 & 0 & 0 & 0 & F2^{(1,0)} \\ 0 & -1 & -1 & -1 & 1 & 1 \end{pmatrix}$$

The vector term is:

```
vec = LinSystem[[1]];
vec // MatrixForm
```

$$\begin{pmatrix} 0 \\ 0 \\ 0 \\ -p0\,F1^{(0,1)} \\ -p0\,F2^{(0,1)} \\ 0 \end{pmatrix}$$

We can use LinearSolve to solve this system, but the result is not pretty. Delete the apostrophe in the next command to see the result.

```
LinearSolve[jac, vec];
```

General problem

In general
(1) There will be many goods, making the di's and p's vectors.
(2) The diagonal terms involving Ui will be blocks corresponding to Jacobians of equations, one block for each demander.
(3) Similarly for the blocks with Fi

Functional forms

We will want to solve such systems with thousands of demanders and hundreds of goods. This is quite feasible.

Typically, the functional forms of the Ui's are the same across people; the differences are really only in the parameter values. For example, Ui could be

$\text{Ui}[q,\ p] = A_i\ (a_i + b_i\ q^{\rho_i})^{\gamma_i}$

Similarly for the production equation, Fi.

For multiple goods, it is often impossible to solve out for the demand vector explicitly; must use implicit form. Same for the output vectors for firms.

So, even if you have thousands of demanders and hundreds of goods, applying AD to construct the derivatives needed to construct LinSystem (which is done only when one has substituted in the parameters for Ui's and Fi's) is feasible.

■ Higher-order terms

We can go to higher-order derivatives

```
eqnstt = D[eqns, τ, τ]; eqns0 = eqnstt /. τ → 0; eqns0 = eqns0 /. sbs;
vars = {P''[0], D1''[0], D2''[0], D3''[0], Q1''[0], Q2''[0]} /. sbs;
LinSystem = CoefficientArrays[eqns0, vars];
```

The solvability matrix is unchanged.

```
LinSystem[[2]] // MatrixForm
```

$$\begin{pmatrix} U1^{(0,1)} & U1^{(1,0)} & 0 & 0 & 0 & 0 \\ U2^{(0,1)} & 0 & U2^{(1,0)} & 0 & 0 & 0 \\ U3^{(0,1)} & 0 & 0 & U3^{(1,0)} & 0 & 0 \\ F1^{(0,1)} & 0 & 0 & 0 & F1^{(1,0)} & 0 \\ F2^{(0,1)} & 0 & 0 & 0 & 0 & F2^{(1,0)} \\ 0 & -1 & -1 & -1 & 1 & 1 \end{pmatrix}$$

The vector term is more complex, but still fits our framework.